New Recycling Cavity calculations

This notebook calculates various parameters of the new recycling cavities.

Preparation

```
In[1]:= << ComplexUtils`
In[2]:= << GaussianOptics`
In[3]:= << PlotLegends`
```

■ Parameters, constants

In[4]:= `ppm = 10`$^{-6}$`; MHz = 10`^{6}`; mm = 10`$^{-3}$`; nm = 10`$^{-9}$`;`

```
In[5]:= SpeedOfLight = {c -> 299 792 458};
In[6]:= waveLength = {λ → 1064 nm};
In[7]:= substrate = {n → 1.45};
```

Cavity Lengths with 11&55MHz modulations

It seems that the 9MHz−45MHz scheme requires a long recycling cavities which are basically impossible to house in our chambers.
Here are cavity lengths with alternative modulation frequencies, 11MHz−55MHz

■ Parameters

In[8]:= `mirrorParams40m = {rp →` $\sqrt{1-0.055}$ `, tp →` $\sqrt{0.055}$ `, rx →` $\sqrt{1-0.014}$ `, ry →` $\sqrt{1-0.014}$ `, rs →` $\sqrt{1-0.1}$ `, ts →` $\sqrt{0.1}$ `, ri →` $\sqrt{1-0.014}$ `, ti →` $\sqrt{0.014}$ `};`

```
In[9]:= baseFreq = 11 MHz;

In[10]:= sb1Frequency = {ω → 2 π baseFreq, f → baseFreq, ω 1-> 2 π baseFreq, f1 → baseFreq};

In[11]:= sb1uFrequency = {ω → (ω /. sb1Frequency),
    ω 1→ (ω /. sb1Frequency), f → (f /. sb1Frequency), f1 -> (f /. sb1Frequency)};

In[12]:= sb1lFrequency = {ω → (-ω /. sb1Frequency),
    ω 1→ (-ω /. sb1Frequency), f → (-f /. sb1Frequency), f1 -> (-f /. sb1Frequency)};

In[13]:= sb2Frequency = {ω → (5 ω /. sb1Frequency),
    ω 2-> (5 ω /. sb1Frequency), f → (5 f /. sb1Frequency), f2 -> (5 f /. sb1Frequency)};

In[14]:= sb2uFrequency = {ω → (5 ω /. sb1Frequency),
    ω 2-> (5 ω /. sb1Frequency), f → (5 f /. sb1Frequency), f2 -> (5 f /. sb1Frequency)};

In[15]:= sb2lFrequency = {ω → (-5 ω /. sb1Frequency), ω 2-> (-5 ω /. sb1Frequency),
    f → (-5 f /. sb1Frequency), f2 -> (-5 f /. sb1Frequency)};
```

Resonant conditions

We determine the macroscopic lengths of the recycling cavities by considering the resonant conditions of the carrier and the sidebands to the cavities.

PRC

PRC is resonant for both sidebands but anti−resonant for the carrier when the arm is not locked.
In order to make the PRC length as short as possible, the FSR of the PRC has to be twice the frequency of SB1. Then SB2 must be odd multiple of SB1.

ν_{prc} is the resonant tune of the PRC for carrier. It represents the phase change of the carrier along the one way of the PRC.
We disregard a large integral part of ν_{prc}. $\nu_{prc}=1$ means the phase change of 2π.

```
In[16]:= PRCTune = Solve[2 ν prc == 0.5, ν prc][[1]]

Out[16]= {ν prc → 0.25}
```

The macroscopic length of the PRC determines the shift of the resonant tunes for sidebands with respect to the carrier.
In order to resonate the SB1 in the PRC, this shift has to be 0.5 so that the total round−trip resonant tune for the SB1 is 1 in the PRC.

```
In[17]:= PRCMacroL = Solve[2 f1 lprc / c == 0.5, lprc][[1]] /. sb1Frequency /. SpeedOfLight

Out[17]= {lprc → 6.81346}
```

SRC

The SRC is resonant for the SB2 and the carrier.
ν_{src} is the resonant tune of the carrier for one way trip of the SRC.

```
In[18]:= SRCdeTune = Solve[2 ν src == 0 / 360, ν src][[1]]

Out[18]= {ν src → 0}
```

The SRC macro length is chosen so that the total resonant tune of SB2 is an integral number for an SRC round−trip.
This integral number is arbitrary, but 2 is chosen to make SB1 as anti−resonant as possible to the SRC.

```
In[19]:= SRCMacroL =
          Solve[2 ν src + 2 f2 lsrc / c == 2, lsrc][[1]] /. SRCdeTune /. SpeedOfLight /. sb2Frequency // N

Out[19]= {lsrc → 5.45077}
```

The resonant tune of SB1 for an SRC round−trip.

```
In[20]:= 2 f1 lsrc / c /. SRCMacroL /. SRCdeTune /. SpeedOfLight /. sb1Frequency

Out[20]= 0.4
```

This is close to the anti−resonance (0.5).

Asymmetry

Schnupp asymmetry is chosen to make the SB2 critically coupled to the dual recycled michelson.

The MI reflectivity and transmissivity can be written as functions of asymmetry lm.

```
In[21]:= rmi = ri Cos[lm ω 2/ c];
```

In[22]:=
```
tmi = ri Sin[lm ω 2/ c];
```

The reflectivity of the compound mirror formed by the MI and the SRM

In[23]:=
$$rcomp = -rmi + \frac{tmi^2\, rs}{1 - rmi\, rs};$$

rcomp should be equal to rp to be critically coupled.

In[24]:=
```
schnuppAsymmetry =
  FindRoot[-rcomp == rp /. mirrorParams40m /. sb2Frequency /. SpeedOfLight, {lm, 0.04}]
```

Out[24]=
```
{lm → 0.0308879}
```

Imperial numbers

Since the 40m optical layout CAD drawing is in imperial units, here we show important lengths in inches.

PRC Length

In[25]:=
```
(lprc /. PRCMacroL) / 0.0254
```

Out[25]=
```
268.247
```

SRC Length

In[26]:=
```
(lsrc /. SRCMacroL) / 0.0254
```

Out[26]=
```
214.597
```

Asymmetry

In[27]:=
```
(lm /. schnuppAsymmetry) / 0.0254
```

Out[27]=
```
1.21606
```

Reflectivities and losses

PRC critical coupling condition

Here we examine how much loss is allowed for the folding mirrors for the recycling cavities.

The reflectivity of the arm cavity

In[28]:=
$$rarm = -ri + \frac{ti^2 \sqrt{1 - RTL}}{1 - ri \sqrt{1 - RTL}};$$

The reflectivity of the PRM for critical coupling is calculated here.
Ltt is the loss of the tip−tilt mirror. Note that there are two TT−stages and the beam sees each tip−tilt mirror twice for a round trip.

In[29]:=
$$rprm = rarm * \left(\sqrt{1 - Ltt}\right)^4;$$

Find the allowed loss for the tip−tilt mirrors given the design reflectivity of the PRM and the round−trip loss of the arms.

In[30]:=
```
Ltt / ppm /. FindRoot[rprm - rp /. mirrorParams40m /. RTL → 180 ppm, {Ltt, 1000 ppm}]
```

Out[30]= 1372.82

Beam Propagation and mode matching

In this section, we deal with the propagation of the beam to decide the radius of curvature (ROC) of the recycling mirrors.

We also attempt to mode match the input beam to the IFO and the output beam to the OMC.

Throughout the section, I will extensively use functions from a *Mathematica* package GaussianOptics‘. This package was written by me to provide combinient functions to calculate the propagation of a Gaussian beam. The documentation of the package can be found in the package file itself. There is also a tutorial notebook on GaussianOptics included in the distribution archive of this file.

Cavity Mode Equation

Just a reminder of basic cavity mode calculations. The results will be used to get the cavity modes of the arm, recycling and mode cleaner cavities.

Beam parameter at the waist : q0

In[31]:= $q0 = i \frac{\pi\, w0^2}{\lambda};$

At the front mirror

In[32]:= $q1 = q0 - d1;$

Require the ROC of the wave and the mirror match

In[33]:= $eq1 = R1 == \mathrm{Simplify}\left[1 \Big/ \mathrm{RealPart}\left[\frac{1}{q1}\right]\right];$

At the end mirror

In[34]:= $q2 = q0 + (L - d1);$

In[35]:= $eq2 = R2 == \mathrm{Simplify}\left[1 \Big/ \mathrm{RealPart}\left[\frac{1}{q2}\right]\right];$

Solve the cavity equations

In[36]:=
```
Solutions = Simplify[Solve[{eq1, eq2}, {d1, w0}]]
```

Out[36]=
$$\left\{\left\{d1 \to \frac{L\,(L-R2)}{2\,L+R1-R2},\ w0 \to -\frac{L^{1/4}\,(-(L+R1)\,(L-R2)\,(L+R1-R2))^{1/4}\,\sqrt{\lambda}}{\sqrt{\pi}\,\left((2\,L+R1-R2)^2\right)^{1/4}}\right\},\right.$$
$$\left\{d1 \to \frac{L\,(L-R2)}{2\,L+R1-R2},\ w0 \to -\frac{i\,L^{1/4}\,(-(L+R1)\,(L-R2)\,(L+R1-R2))^{1/4}\,\sqrt{\lambda}}{\sqrt{\pi}\,\left((2\,L+R1-R2)^2\right)^{1/4}}\right\},$$
$$\left\{d1 \to \frac{L\,(L-R2)}{2\,L+R1-R2},\ w0 \to \frac{i\,L^{1/4}\,(-(L+R1)\,(L-R2)\,(L+R1-R2))^{1/4}\,\sqrt{\lambda}}{\sqrt{\pi}\,\left((2\,L+R1-R2)^2\right)^{1/4}}\right\},$$
$$\left.\left\{d1 \to \frac{L\,(L-R2)}{2\,L+R1-R2},\ w0 \to \frac{L^{1/4}\,(-(L+R1)\,(L-R2)\,(L+R1-R2))^{1/4}\,\sqrt{\lambda}}{\sqrt{\pi}\,\left((2\,L+R1-R2)^2\right)^{1/4}}\right\}\right\}$$

Pick the physical solution

In[37]:=
```
cavityEigenMode = Solutions[[4]];
```

In[38]:=
```
waistPosition = d1 /. cavityEigenMode
```

Out[38]=
$$\frac{L\,(L-R2)}{2\,L+R1-R2}$$

In[39]:=
```
waistSize = w0 /. cavityEigenMode
```

Out[39]=
$$\frac{L^{1/4}\,(-(L+R1)\,(L-R2)\,(L+R1-R2))^{1/4}\,\sqrt{\lambda}}{\sqrt{\pi}\,\left((2\,L+R1-R2)^2\right)^{1/4}}$$

- **40m ArmCavity**

Eigen mode of the 40m arm cavity

In[40]:=
```
armCavityParams = {L → 38, R1 → -10000, R2 → 57.37, λ → 1064 10^-9};
```

In[41]:=
```
waistPosition /. armCavityParams // N
```

Out[41]= 0.0737434

In[42]:=
```
waistSize /. armCavityParams // N
```

Out[42]= 0.00303267

q parameter at the waist of the arm cavity (at the ITM)

In[43]:=
```
q0Arm = q0 /. cavityEigenMode /. armCavityParams // N
```

Out[43]= $0. + 27.1556\,i$

PRC

Optical Length and Gaussian Length

Optical Length is the length used for calculating the resonant conditions. For a geometric length L, the optical length is n*L, where n is the index of refraction.
Gaussian Length is the effective length to be used for calculations of Gaussian beam propagation.
For Gaussian beams, a geometric distance L looks like L/n in a medium with index of refraction n.

```
In[44]:= optLprc = lprc /. PRCMacroL
Out[44]= 6.81346
```

Actual optical lengths for each arms taking into account the Schnupp asymmetry.
Xarm is shorter.

```
In[45]:= optLprcX = (optLprc - lm / 2 /. schnuppAsymmetry);
In[46]:= optLprcY = (optLprc + lm / 2 /. schnuppAsymmetry);
```

Convert to the Gaussian length.
Take into account the fused silica path in the ITM substrate and the BS. For fused silica, n=1.45.

```
In[47]:= LprcXG = optLprcX - (1.45 - 1 / 1.45) * (Litm + Lbs) /. {Litm → 25.4 mm, Lbs → 1.14666 * 25.4 mm};
In[48]:= LprcYG = optLprcY - (1.45 - 1 / 1.45) * Litm /. {Litm → 25.4 mm, Lbs → 1.14666 * 25.4 mm};
```

Average Gaussian length of the PRC

In[49]:= $\text{LprcG} = \frac{\text{LprcXG} + \text{LprcYG}}{2};$

PRM ROC

Propagate the beam from the arm cavity waist (on the ITM HR surface) to the PRM position.

```
In[50]:= qPRM = CompileBeamPath[{FreeSpace[LprcG]}, q0Arm][LprcG]
Out[50]= 6.78308 + 27.1556 i
```

Convert the q−parameter into the radius of curvature.

```
In[51]:= Rprm0 = QtoROC[qPRM, λ /. waveLength]
Out[51]= 115.499
```

The beam size on the PRM

```
In[52]:= QtoBeamSize[qPRM, λ /. waveLength]
Out[52]= 0.00312585
```

ABCD matrix for a round trip in PRC

In[53]:= $\text{Tprc} = \text{Simplify}\left[\begin{pmatrix} 1 & \text{LprcG} \\ 0 & 1 \end{pmatrix} . \begin{pmatrix} 1 & 0 \\ \frac{-2}{\text{Rprm}} & 1 \end{pmatrix} . \begin{pmatrix} 1 & \text{LprcG} \\ 0 & 1 \end{pmatrix}\right];$

g factor

In[54]:=
```
gprc = Tr[Tprc] / 2;
```

In[55]:=
```
Manipulate[Row[{"g = ", gprc}] /. Rprm → x, {{x, Rprm0, "PRM ROC"}, 30, 200}]
```

Out[55]=
```
PRM ROC
115.499
g = 0.882543
```

- **Transverse Mode Interval**

In[56]:=
```
γprc= ArcCos[gprc] / π;
```

In[57]:=
```
Manipulate[Row[{"γ = ", γprc}] /. Rprm → x, {{x, Rprm0, "PRM ROC"}, 30, 200}]
```

Out[57]=
```
PRM ROC
115.499
γ = 0.15583
```

γ is 0.155737 when the ROC of PRM is 115.633. It becomes 0.2 at ROC= 71 and becomes 0.15 at ROC=124.5.

- **PRC cavity mode**

The waist of the PRC should be located at the ITM

In[58]:=
```
q0PRC = q0 /. cavityEigenMode /. {L → LprcG, R1 → -Rprm, R2 → 10 000};
```

Mode matching between the arm mode and the PRC mode.

In[59]:=
```
Manipulate[Row[{"Mode matching rate = ", Abs[ModeMatchingCoeff[q0Arm, q0PRC /. Rprm → x]]}],
 {{x, Rprm0, "PRM ROC"}, 60, 200}]
```

Out[59]=
```
PRM ROC
115.499
Mode matching rate = 0.999997
```

In[60]:=
```
Plot[ModeMatchingCoeff[q0Arm, q0PRC],
 {Rprm, 60, 200}, AxesLabel → {"PRM ROC", "Mode matching rate"},
 GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[60]=



- **Beam propagation**

The ABC matrix represents the lens effect of the PRM substrate.

In[61]:=
```
Manipulate[PlotBeamPropagation[{FreeSpace[LprcG], ABCDMatrix[{{1, 0}, {(1.45 - 1) / Rprm, 1}}]},
  q0Arm, λ /. waveLength, 20], {{Rprm, Rprm0}, 60, 200}]
```

Out[61]=



- ## SRC

- ### Geometric Length, Optical Length and Gaussian Length

Optical Length is the length used for calculating the resonant conditions. Gaussian Length is the effective length for Gaussian beam propagation

In[62]:=
```
optLsrc = lsrc /. SRCMacroL
```

Out[62]= 5.45077

Actual optical lengths for each arms.
Xarm is shorter

In[63]:=
```
optLsrcX = (optLsrc - lm / 2 /. schnuppAsymmetry);
```

In[64]:=
```
optLsrcY = (optLsrc + lm / 2 /. schnuppAsymmetry);
```

Convert to the Gaussian length

```
In[65]:= LsrcXG = optLsrcX - (1.45 - 1 / 1.45) * (Litm + Lbs * 2) /. {Litm → 25.4 mm, Lbs → 1.14666 * 25.4 mm};
```

```
In[66]:= LsrcYG = optLsrcY - (1.45 - 1 / 1.45) * (Litm + Lbs) /. {Litm → 25.4 mm, Lbs → 1.14666 * 25.4 mm};
```

Average Gaussian length of the SRC

In[67]:= $LsrcG = \frac{LsrcXG + LsrcYG}{2}$

```
Out[67]= 5.39824
```

- **SRM ROC**

```
In[68]:= qSRM = CompileBeamPath[{FreeSpace[LsrcG]}, q0Arm][LsrcG]
```

```
Out[68]= 5.39824 + 27.1556 i
```

The ROC of SRM

```
In[69]:= Rsrm0 = QtoROC[qSRM, λ /. waveLength]
```

```
Out[69]= 142.004
```

Beam size on the SRM

```
In[70]:= QtoBeamSize[qSRM, λ /. waveLength]
```

```
Out[70]= 0.00309202
```

- **ABCD matrix for a round trip in SRC**

In[71]:= $Tsrc = Simplify\left[\begin{pmatrix} 1 & LsrcG \\ 0 & 1 \end{pmatrix} . \begin{pmatrix} 1 & 0 \\ \frac{-2}{Rsrm} & 1 \end{pmatrix} . \begin{pmatrix} 1 & LsrcG \\ 0 & 1 \end{pmatrix}\right];$

- **g factor**

```
In[72]:= gsrc = Tr[Tsrc] / 2;
```

```
In[73]:= Manipulate[Row[{"g = ", gsrc}] /. Rsrm → x, {{x, Rsrm0, "SRM ROC"}, 30, 200}]
```

```
Out[73]=
SRM ROC
142.004
g = 0.92397
```

- **Transverse Mode Interval**

```
In[74]:= γ src= ArcCos[gsrc] / π;
```

In[75]:=
```
Manipulate[Row[{"γ = ", γ src}] /. Rsrm → x, {{x, Rsrm0, "SRM ROC"}, 30, 200}]
```

Out[75]=



γ is 0.124849 when the ROC of PRM is 142.172. It becomes 0.2 at ROC= 56 and becomes 0.15 at ROC=99.

- **SRC cavity mode**

The waist of the SRC should be located on the ITM.

In[76]:=
```
q0SRC = q0 /. cavityEigenMode /. {L → LsrcG, R1 → -Rsrm, R2 → 10 000};
```

Mode matching between the arm mode and the SRC mode.

In[77]:=
```
Manipulate[Row[{"Mode matching rate = ", Abs[ModeMatchingCoeff[q0Arm, q0SRC /. Rsrm → x]]}],
 {{x, Rsrm0, "SRM ROC"}, 60, 200}]
```

Out[77]=



In[78]:=
```
Plot[ModeMatchingCoeff[q0Arm, q0SRC],
 {Rsrm, 100, 220}, AxesLabel → {"SRM ROC", "Mode matching rate"},
 GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[78]=



- **Beam propagation**

The ABC matrix represents the lens effect of the SRM substrate.

In[79]:=
```
Manipulate[PlotBeamPropagation[{FreeSpace[LsrcG], ABCDMatrix[{{1, 0}, {(1.45 - 1) / Rsrm, 1}}]},
  q0Arm, λ /. waveLength, 20], {{Rsrm, Rsrm0}, 100, 200}]
```

Out[79]=



- ## Input Mode Matching

- ### MC output mode

In[80]:=
```
MCParam = {R1 → 100 000, R2 → 18.4,  L → 13.6269};
```

q−parameter of the beam at the MC waist.

In[81]:=
```
q0MC = q0 /. cavityEigenMode /. MCParam // N
```

Out[81]= 0. + 8.06454 i

- ### MMT parameters

d1 is the distance from the MC waist to the MMT1. dmmt is the distance between the two MMT mirrors. d2 is the distance from the MMT2 to the PRM front surface. fmmt1 and fmmt2 are the focal lengths of MMT1 and MMT2.

These are the current MMT parameters.

In[82]:=
```
MMTParam0 = {d1 → 1310 mm, dmmt → 137 mm, d2 → 1366 mm, fmmt1 → -168 mm, fmmt2 → 687.5 mm};
```

- **Propagate the beam from the MC waist to the PRM.**

Beam path object

In[83]:=
```
InputBeamPath = {FreeSpace[d1], Lens[fmmt1], FreeSpace[dmmt],
    Lens[fmmt2], FreeSpace[d2], ABCDMatrix[{{1, 0}, {(1.45 - 1) / Rprm0, 1}}]};
```

In[84]:=
```
CompiledInputBP = CompileBeamPath[InputBeamPath, q0MC];
```

- **Beam profile to be matched**

The input beam q−parameter should match the q−parameter of the cavity eigen mode on the PRM, qPRM.

In[85]:=
```
qPRM = CompileBeamPath[{FreeSpace[LprcG]}, q0Arm][-LprcG]
```

Out[85]= -6.78308 + 27.1556 i

- **See if the new mode matching can be achieved by just moving the distance between the MMT mirrors (dmmt).**

In[86]:=
```
Plot[Abs[ModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParam0]],
  {x, 100 mm, 200 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
  GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[86]=



It is impossible to achieve a good mode matching just by moving dmmt.

- **Find the optimal mode matching**

Now we will change the focal lengths of the MMT mirrors.
Scan the focal lengths of the MMT mirrors to find the optimal focal length pair.

In[87]:=
```
Ans1 = FindMaximum[Abs[ModeMatchingCoeff[qPRM,
     CompiledInputBP[d1 + dmmt + d2] /. {fmmt1 → x, fmmt2 → y} /. MMTParam0]],
   {{x, -400 mm}, {y, 300 mm}}, AccuracyGoal → 5, PrecisionGoal → 5]
```

Out[87]= {1., {x → -0.154225, y → 0.28988}}

Looks like the focal length of MMT1 does not have to be changed much. So we will try to find a good mode maching condition by changing dmmt and fmmt2.

In[88]:=
```
Ans2 = FindMaximum[Abs[ModeMatchingCoeff[qPRM,
     CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. fmmt2 → y /. MMTParam0]],
   {{x, 137 mm}, {y, 200 mm}}, AccuracyGoal → 3, PrecisionGoal → 3]
```

Out[88]= {1., {x → 0.149162, y → 0.315565}}

The optimal MMT parameters

According to the above optimization, we should change the distance between the MMT mirrors from 137mm to 149.2mm and the focal length of MMT2 from 687.5mm to 315.6mm.

In[89]:=
```
MMTParam = Join[{dmmt → x, fmmt2 → y} /. Ans2[[2]], MMTParam0[[{1, 3, 4}]]] // N
```

Out[89]=
```
{dmmt → 0.149162, fmmt2 → 0.315565, d1 → 1.31, d2 → 1.366, fmmt1 → -0.168}
```

- **Plot the contour of mode matching rate in the plane of dmmt and fmmt2.**

In[90]:=
```
Plot3D[Abs[
   ModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. fmmt2 → y /. MMTParam]],
  {x, 100 mm, 200 mm}, {y, 250 mm, 350 mm}, PlotRange → All, AxesLabel → {"dmmt", "fmmt2"}]
```

Out[90]=



- **Propagation of the beam from the MC waist to the PRM surface**

In[91]:=
```
PlotBeamPropagation[InputBeamPath /. MMTParam, q0MC, λ /. waveLength, d1 + dmmt + d2 /. MMTParam]
```

Out[91]=



Output Mode Matching

- **OMC output mode**

In[92]:=
```
OMCParam = {R1 → 1 000 000, R2 → 1, L → 238 mm};
```

q−parameter of the beam at the MC waist.

In[93]:=
```
q0OMC = q0 /. cavityEigenMode /. OMCParam // N
```

Out[93]=
```
0. + 0.425859 i
```

- **OMMT parameters**

d1 is the distance from the SRM to the OMMT1. dmmt is the distance between the two OMMT mirrors. d2 is the distance from the OMMT2 to the waist of the OMC. fmmt1 and fmmt2 are the focal lengths of OMMT1 and OMMT2.

These are the current OMMT parameters.

```
In[94]:= OMMTParam0 = {d1 → 5140 mm, dmmt → 384 mm, d2 → 809 mm, fmmt1 → 618.4 / 2 mm, fmmt2 → 150 / 2 mm};
```

- **q−parameter on the SRM**

```
In[95]:= qSRM = CompileBeamPath[{FreeSpace[LsrcG]}, q0Arm][LsrcG]

Out[95]= 5.39824 + 27.1556 i
```

- **Propagate the beam from SRM to the OMC waist.**

We propagate the q parameter on the SRM (qSRM) through the SRM substrate, the OMMT to the OMC.

```
In[96]:= OutputBeamPath = {ABCDMatrix[{{1, 0}, {(1.45 - 1) / Rsrm0, 1}}],
            FreeSpace[d1], Lens[fmmt1], FreeSpace[dmmt], Lens[fmmt2], FreeSpace[d2]};

In[97]:= CompiledOutputBP = CompileBeamPath[InputBeamPath, qSRM];
```

- **See if the new mode matching can be achieved by just moving the distance between the OMMT mirrors (dmmt).**

```
In[98]:= Plot[Abs[ModeMatchingCoeff[q0OMC, CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParam0]],
          {x, 350 mm, 450 mm}, AxesLabel → {"dmmt", "Mode matching rate"}, PlotRange → All,
          GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[98]=



Looks like we can get a good mode matching rate with a tweak of dmmt.

- **Find the optimal mode matching**

Now we find the optimal distance between OMMT1 and 2.

```
In[99]:= Ans1 = FindMaximum[
           Abs[ModeMatchingCoeff[q0OMC, CompiledOutputBP[d1 + dmmt + d2] /. {dmmt → x} /. OMMTParam0]],
           {{x, 400 mm}}, AccuracyGoal → 5, PrecisionGoal → 5]

Out[99]= {0.999279, {x → 0.391095}}
```

By changing the dmmt from 384mm to 391mm, we can achieve more than 99.9% mode matching.

This is a zoomed plot of the mode matching rate around dmmt=391mm

In[100]:=
```
Plot[Abs[ModeMatchingCoeff[q0OMC, CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParam0]],
  {x, 385 mm, 395 mm}, AxesLabel → {"dmmt", "Mode matching rate"}, PlotRange → All,
  GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[100]=



According to the above optimization, we should change the distance between the OMMT mirrors from 384mm to 391mm.
To achieve more than 95% mode matching, the accuracy of the distance should be +/−2mm.

In[101]:=
```
OMMTParam = Join[{dmmt → x} /. Ans1[[2]], OMMTParam0[[{1, 3, 4, 5}]]] // N
```

Out[101]=
```
{dmmt → 0.391095, d1 → 5.14, d2 → 0.809, fmmt1 → 0.3092, fmmt2 → 0.075}
```

Now just out of curiosity, check what is the optimal mode matching if we allow to change the focal length of OMMT1 too.

In[102]:=
```
FindMaximum[Abs[ModeMatchingCoeff[q0OMC,
    CompiledOutputBP[d1 + dmmt + d2] /. {dmmt → x, fmmt1 → y} /. OMMTParam0]],
  {{x, 400 mm}, {y, 600 mm}}, AccuracyGoal → 4, PrecisionGoal → 4]
```

Out[102]=
```
{1., {x → 0.403076, y → 0.321091}}
```

- **Propagation of the beam from SRM to the waist of OMC**

```
In[103]:= PlotBeamPropagation[OutputBeamPath /. OMMTParam,
  qSRM, λ /. waveLength, d1 + dmmt + d2 /. OMMTParam]
```

Out[103]=



■ Do we need off−axis parabolic mirrors ?

Here we will estimate the astigmatism caused by an off−axis incidence to the mode matching mirrors when we use spherical mirrors.

- **Input MMT**

θ_1 is the incident angle to the MMT1, θ_2 is the incident angle to the MMT2.

```
In[104]:= OffAxisParamI = {θ1 → 17.531 / 2 * π / 180, θ2 → 17.531 / 2 * π / 180 };
```

When a beam is incident at an angle θ to a spherical mirror, the focal length is changed by a factor cos[θ] in the plane of inci−dence, whereas in the perpendicular plane the factor is 1/cos[θ].

```
In[105]:= MMTParamH = Join[{fmmt1 → (fmmt1 Cos[θ1] /. MMTParam /. OffAxisParamI),
    fmmt2 -> (fmmt2 Cos[θ2] /. MMTParam /. OffAxisParamI)}, MMTParam[[{1, 3, 4}]]]

Out[105]= {fmmt1 → −0.166038, fmmt2 → 0.311879, dmmt → 0.149162, d1 → 1.31, d2 → 1.366}
```

```
In[106]:= MMTParamV = Join[{fmmt1 → (fmmt1 / Cos[θ1] /. MMTParam /. OffAxisParamI),
    fmmt2 -> (fmmt2 / Cos[θ2] /. MMTParam /. OffAxisParamI)}, MMTParam[[{1, 3, 4}]]]

Out[106]= {fmmt1 → −0.169985, fmmt2 → 0.319294, dmmt → 0.149162, d1 → 1.31, d2 → 1.366}
```

Plot the mode matching rate of horizontal and vertical directions respectively.

In[107]:=
```
Plot[{Abs[ModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamH]],
      Abs[ModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamV]]},
  {x, 135 mm, 160 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
  GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[107]=



Total mode matching rate of the astigmatic beam into the arm cavity

In[108]:=
```
Plot[
  Abs[AstigmaticModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamH,
    CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamV]]
       , {x, 135 mm, 160 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
  GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[108]=



Find the optimal dmmt

In[109]:=
```
FindMaximum[
  Abs[AstigmaticModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamH,
    CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamV]], {{x, 145 mm}}]
```

Out[109]=
```
{0.936005, {x → 0.149084}}
```

Scan also the focal length of MMT2 to see if we can find a better mode matching rate.

In[110]:=
```
FindMaximum[Abs[AstigmaticModeMatchingCoeff[qPRM,
    CompiledInputBP[d1 + dmmt + d2] /. {dmmt → x, fmmt2 → y Cos[θ/2]} /. MMTParamH /. OffAxisParamI,
    CompiledInputBP[d1 + dmmt + d2] /. {dmmt → x, fmmt2 → y / Cos[θ/2]} /. MMTParamV /.
     OffAxisParamI]], {{x, 160.08 mm}, {y, 200 mm}}, AccuracyGoal → 4, PrecisionGoal → 4]
```

Out[110]=
```
{0.940995, {x → 0.126381, y → 0.292982}}
```

This is not much better than the case only dmmt was changed. In addition, dmmt gets shorter than the current setup in this case. This is not what we want because the AS port beam goes through between the two MMT mirrors.

Combination of an OAP and a spherical mirror

If we keep the MMT1 as it is, this mirror is actually an off−axis parabolic (OAP) mirror. So, the astigmatism from this mirror is very small.
To see what happens to the mode matching rate in this case, we set θ 1=0 in OffAxisParamI so that the astigmatism of the first mirror is zero.

```
In[111]:= OffAxisParamI = {θ 1→ 0 / 2 * π / 180, θ 2→ 17.531 / 2 * π / 180 };

In[112]:= MMTParamH = Join[{fmmt1 → (fmmt1 Cos[θ 1] /. MMTParam /. OffAxisParamI),
              fmmt2 -> (fmmt2 Cos[θ 2] /. MMTParam /. OffAxisParamI)}, MMTParam[[{1, 3, 4}]]]

Out[112]= {fmmt1 → -0.168, fmmt2 → 0.311879, dmmt → 0.149162, d1 → 1.31, d2 → 1.366}

In[113]:= MMTParamV = Join[{fmmt1 → (fmmt1 / Cos[θ 1] /. MMTParam /. OffAxisParamI),
              fmmt2 -> (fmmt2 / Cos[θ 2] /. MMTParam /. OffAxisParamI)}, MMTParam[[{1, 3, 4}]]]

Out[113]= {fmmt1 → -0.168, fmmt2 → 0.319294, dmmt → 0.149162, d1 → 1.31, d2 → 1.366}
```

Plot of mode matching rate for horizontal and vertical directions respectively.

```
In[114]:= Plot[{Tooltip[Abs[ModeMatchingCoeff[qPRM,
                CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamH]], "Horizontal"],
                  Tooltip[Abs[ModeMatchingCoeff[qPRM,
                CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamV]], "Vertical"]},
            {x, 135 mm, 160 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
            GridLines → Automatic,
            GridLinesStyle → Directive[Brown, Dashed]]
```

Out[114]=



Total mode matching rate

In[115]:=
```
Plot[
 Abs[AstigmaticModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamH,
   CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamV]]
     , {x, 135 mm, 160 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
 GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[115]=



Scan dmmt and fmmt2 to find the best mode matching.

In[116]:=
```
FindMaximum[Abs[AstigmaticModeMatchingCoeff[qPRM,
   CompiledInputBP[d1 + dmmt + d2] /. {dmmt → x, fmmt2 → y Cos[θ2]} /. MMTParamH /. OffAxisParamI,
   CompiledInputBP[d1 + dmmt + d2] /. {dmmt → x, fmmt2 → y / Cos[θ2]} /. MMTParamV /.
    OffAxisParamI]], {{x, 160.08 mm}, {y, 200 mm}}, AccuracyGoal → 4, PrecisionGoal → 4]
```

Out[116]=
```
{0.767191, {x → 0.148859, y → 0.315367}}
```

Surprisingly, the maximam mode matching is only 77% in this case. The reason behind this is the following. When both of the MMT mirrors have astigmatism, the effects somewhat cancel out each other and provide a better mode matching than the case only one mirror has astigmatism.
Therefore, we should not combine an OAP and a spherical mirror.

The following interactive plot shows how the horizontal and vertical beam profiles change as we change the incident angle $\theta 1$ and $\theta 2$ as well as dmmt.

In[117]:=
```
Manipulate[
 Plot[{QtoBeamSize[CompiledInputBP[z], λ] /. waveLength /. {dmmt → d, fmmt1 → Cos[θ 1] fmmt1,
      fmmt2 → Cos[θ 2] fmmt2} /. MMTParam, QtoBeamSize[CompiledInputBP[z], λ] /. waveLength /.
     {dmmt → d, fmmt1 → fmmt1 / Cos[θ 1] , fmmt2 → fmmt2 / Cos[θ 2] } /. MMTParam},
  {z, 2, 25}, PlotRange → All], {{θ 1, 0}, 0, 0.15}, {{θ 2, 0}, 0, 0.15},
 {{d, 149 mm, "dmmt"}, 140 mm, 155 mm}]
```

Out[117]=



■ **Longer input MMT**

Since the input MMT suffers from large astigmatism, it is a good idea to elongate it to make the incident angles smaller.
If we increase the MMT length by 122mm, the incident angle becomes 4.7deg. We will find a suitable focal lengths for this

In[130]:=
```
OffAxisParamI = {θ 1→ 4.7 * π / 180, θ 2→ 4.7 * π / 180 };
```

In[139]:=
```
MMTParamH = {fmmt1 → (fmmt1 Cos[θ 1] /. MMTParam /. OffAxisParamI),
   fmmt2 -> (fmmt2 Cos[θ 2] /. MMTParam /. OffAxisParamI), d1 → 1.435, d2 → 1.366, dmmt → 262 mm};
```

In[140]:=
```
MMTParamV = {fmmt1 → (fmmt1 / Cos[θ 1] /. MMTParam /. OffAxisParamI), fmmt2 ->
     (fmmt2 / Cos[θ 2] /. MMTParam /. OffAxisParamI), d1 → 1.435, d2 → 1.366, dmmt → 262 mm};
```

Find the optimal mode matching by changing the focal lengths of MMT1 and MMT2

In[142]:=
```
Ans3 = FindMaximum[Abs[AstigmaticModeMatchingCoeff[qPRM,
     CompiledInputBP[d1 + dmmt + d2] /. {fmmt1 → x Cos[θ 1], fmmt2 → y Cos[θ 2]} /. MMTParamH /.
      OffAxisParamI, CompiledInputBP[d1 + dmmt + d2] /.
        {fmmt1 → x / Cos[θ 1], fmmt2 → y / Cos[θ 2]} /. MMTParamV /. OffAxisParamI]],
   {{x, -160 mm}, {y, 300 mm}}, AccuracyGoal → 4, PrecisionGoal → 4]
```

Out[142]=
```
{0.998197, {x → -0.301278, y → 0.558003}}
```

The best parameters for the longer MMT.

```
In[149]:= MMTParamL = Join[{fmmt1 → x, fmmt2 → y} /. Ans3[[2]], MMTParamH[[{3, 4, 5}]]] // N

Out[149]= {fmmt1 → -0.301278, fmmt2 → 0.558003, d1 → 1.435, d2 → 1.366, dmmt → 0.262}

MMTParamHL = Join[{fmmt1 → x Cos[θ1], fmmt2 → y Cos[θ2]} /. Ans3[[2]] /. OffAxisParamI,
    MMTParamL[[{3, 4, 5}]]];

In[148]:= MMTParamVL = Join[{fmmt1 → x / Cos[θ1], fmmt2 → y / Cos[θ2]} /. Ans3[[2]] /. OffAxisParamI,
    MMTParamL[[{3, 4, 5}]]];
```

Total mode matching rate as a function of dmmt

```
In[152]:= Plot[Abs[
    AstigmaticModeMatchingCoeff[qPRM, CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamHL,
     CompiledInputBP[d1 + dmmt + d2] /. dmmt → x /. MMTParamVL]]
       , {x, 255 mm, 269 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
   GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[152]=



- **Output MMT**

θ1 is the incident angle to the MMT1, θ2 is the incident angle to the MMT2.

```
In[119]:= OffAxisParamO = {θ1→ 5.935 / 2 * π / 180, θ2→ 5.935 / 2 * π / 180 };
```

When a beam is incident at an angle θ to a spherical mirror, the focal length is changed by a factor cos[θ] in the plane of incidence, whereas in the perpendicular plane the factor is 1/cos[θ].

```
In[120]:= OMMTParamH = Join[{fmmt1 → (fmmt1 Cos[θ1] /. OMMTParam /. OffAxisParamO),
    fmmt2 -> (fmmt2 Cos[θ2] /. OMMTParam /. OffAxisParamO)}, OMMTParam[[{1, 2, 3}]]]

Out[120]= {fmmt1 → 0.308785, fmmt2 → 0.0748994, dmmt → 0.391095, d1 → 5.14, d2 → 0.809}

In[121]:= OMMTParamV = Join[{fmmt1 → (fmmt1 / Cos[θ1] /. OMMTParam /. OffAxisParamO),
    fmmt2 -> (fmmt2 / Cos[θ2] /. OMMTParam /. OffAxisParamO)}, OMMTParam[[{1, 2, 3}]]]

Out[121]= {fmmt1 → 0.309615, fmmt2 → 0.0751007, dmmt → 0.391095, d1 → 5.14, d2 → 0.809}
```

Plot the mode matching rate of horizontal and vertical directions respectively.

In[122]:=
```
Plot[
 {Abs[ModeMatchingCoeff[q0OMC, CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParamH]],
      Abs[ModeMatchingCoeff[q0OMC,
    CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParamV]]},
 {x, 385 mm, 398 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
 GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[122]=



Total mode matching rate of the astigmatic beam into the OMC cavity

In[123]:=
```
Plot[Abs[
  AstigmaticModeMatchingCoeff[q0OMC, CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParamH,
   CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParamV]]
     , {x, 385 mm, 398 mm}, AxesLabel → {"dmmt", "Mode matching rate"},
 GridLines → Automatic, GridLinesStyle → Directive[Brown, Dashed]]
```

Out[123]=



Find the optimal dmmt

In[124]:=
```
FindMaximum[Abs[
  AstigmaticModeMatchingCoeff[q0OMC, CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParamH,
   CompiledOutputBP[d1 + dmmt + d2] /. dmmt → x /. OMMTParamV]], {{x, 390 mm}}]
```

Out[124]=
```
{0.991665, {x → 0.391092}}
```

We can achieve more than 99% mode matching.

Scan also the focal length of OMMT1 to see if we can find a better mode matching rate.

```
In[125]:= FindMaximum[Abs[AstigmaticModeMatchingCoeff[q0OMC,
    CompiledOutputBP[d1 + dmmt + d2] /. {dmmt → x, fmmt1 → y Cos[θ 2]} /. OMMTParamH /.
     OffAxisParamO, CompiledOutputBP[d1 + dmmt + d2] /. {dmmt → x, fmmt1 → y / Cos[θ 2]} /.
      OMMTParamV /. OffAxisParamO]],
  {{x, 390 mm}, {y, 300 mm}}, AccuracyGoal → 5, PrecisionGoal → 5]

Out[125]= {0.993199, {x → 0.403533, y → 0.321547}}
```

This is not much better than the case only dmmt was changed.